
Repeated Measurements – example 2
The following data come from an experiment on Ladybirds (“nyckelpigor”; an insect). The purpose of the experiment was to investigate whether different treatments had different attraction on ladybirds and whether this changed with time. 

Plots of land were divided into three blocks, each containing ten plots. The plots were treated with one of  the ten treatments. The number of ladybirds was counted on each of the plots after 5, 6, 9, 13, 16 and 21 days.

Analyze these data to find out if different treatments have different attraction on ladybirds and whether this changes with time. 

The variables in the enclosed file are

Obs 
Observation number (not used)

Plot 
Plot number

Treat 
Treatment

Block 
Block

t 
time (5, 6, 9, 13, 16 or 21)

y
Number of ladybirds

Solution in SAS:

The model here is a fixed factor (treatment) in 3 blocks. For each plot we make 6 measurements in time. We start by making a plot over the data: 

proc sgpanel data=ladybirds;

panelby block;

series x=t y=y /group=treat;

run;
We can make some observation in these plots, e.g. we see that there is quite a lot variation at time point 6 (i.e. the different treatments give different results, treatment HA and KF have the highest values in two of the blocks). 

When we set up the model we need to consider that the distance between time points is not equidistant, i.e. sometimes we take measurements on consecutive days sometimes with a time distance with up to 5 days. Hence, we cannot use AR(1) as time series correlation structure and compare it with the UN structure, but would choose a spatial model, e.g. spatial power. The results from the fit with the unstructured covariance matrix is not given here, the fit is almost the same as with the model below:
ods graphics on; 

proc mixed data=ladybirds plots=(all);

class plot treat block t;

model y=treat t treat*t block/ddfm=kr;

*random block;
repeated /subject=plot type=sp(pow)(t);

run;

ods graphics off;

We see that the residuals are almost normally distributed, but maybe too many small and too few high residuals. The first plot looks as there is an increase in the residual variation as the predicted mean increases. We can check if a transformation might be a solution, e.g. using a log-transform. Since some of the y-values are 0 we will use log(y+0.5) as response.

data ladybirds1;

set ladybirds;

logy=log(y+0.5);

run;

proc mixed data=ladybirds1 plots=(all);

class plot treat block t;

model logy=treat t treat*t block/ddfm=kr;

repeated /subject=plot type=sp(pow)(t);

run;
We see that the distribution of the residuals is a little closer to normal. In the first residual plot we can see some ‘lines’ which are effects of the data being whole numbers. The lowest line are the 25 observations with the value 0. The choice of how to handle values with a 0 count can in this case influence our results, e.g. if you choose the transformation log(y+0.1) we get another result and another residual plot. Even if the plots look of and normal distribution could be assumed, it still does not feel really satisfying. 

We can still note the results of our tests, where we can see a significant time effect, but no significant other effects. : 

	Type 3 Tests of Fixed Effects

	Effect
	Num DF
	Den DF
	F Value
	Pr > F

	Treat
	9
	80.5
	0.31
	0.9703

	t
	5
	87.4
	24.69
	<.0001

	Treat*t
	45
	87.6
	0.78
	0.8243

	Block
	2
	85.1
	2.18
	0.1193


If we want to handle this data in a better way without making the result dependent on the choice of transformation we would consider other distribution than normal. For data like this where we monitor whole number (0,1,2,3,…) a good approximation is often the Poisson distribution. Mixed principles work the same way as for normal data, but we move into the area of generalized linear mixed models. For this we need to use the PROC GLIMMIX. 
proc glimmix data=ladybirds1 plots=(all);

class plot treat block t;

model y=treat t treat* t block /dist=poisson ddfm=kr;

random  residual/subject=plot type=sp(pow)(t) ;

run;
The advantage with this approach is that we no longer need to adjust values that are 0. We can see that the results of the tests are quite similar to the tests from the analysis of log-transformed data. 

	Type III Tests of Fixed Effects

	Effect
	Num DF
	Den DF
	F Value
	Pr > F

	Treat
	9
	109.7
	0.40
	0.9303

	Block
	2
	86.12
	2.26
	0.1102

	t
	5
	102.3
	23.33
	<.0001

	Treat*t
	45
	94.5
	0.91
	0.6335


A restriction in the use of Poisson models is that the model assumes that the variance is equal to the mean. We can therefore often observe overdispersion in such models, i.e. the actual observed variation is too large to fit the theoretical model. An indication of overdispersion is that the values of Gener. Chi-Square /DF is (much) larger than 1, which is the case here. 
	Fit Statistics

	-2 Res Log Pseudo-Likelihood
	347.89

	Generalized Chi-Square
	233.66

	Gener. Chi-Square / DF
	2.00


A possibility to solve this would be to use the negative binomial distribution dist=negbin which allows for a variance that is different from the mean. This however does not converge for this data and leaves us with the above model as the final model. 

